

```
/* MM_data.sas */
```

```
data MM_HCD_full;  
input author_index study_index mouse_line $ N_tumors N_mice N_males  
N_females N_male_tumors N_female_tumors;  
datalines;  
1 1 F 0 143 45 98 0 0  
1 2 F 0 100 29 71 0 0  
2 3 F 0 64 0 64 0 0  
3 4 F 0 69 0 69 0 0  
4 5 F 0 234 40 194 0 0  
5 6 C 0 105 53 52 0 0  
5 7 C 0 96 47 49 0 0  
5 8 C 0 100 50 50 0 0  
5 9 C 0 97 49 48 0 0  
5 10 C 0 99 50 49 0 0  
5 11 C 0 119 59 60 0 0  
5 12 C 0 100 50 50 0 0  
5 13 C 0 120 60 60 0 0  
5 14 C 0 100 50 50 0 0  
5 15 C 0 96 48 48 0 0  
5 16 C 0 100 50 50 0 0  
5 17 C 0 99 50 49 0 0  
5 18 C 0 139 69 70 0 0  
5 19 C 0 99 50 49 0 0  
5 20 C 0 118 59 59 0 0  
5 21 C 0 120 60 60 0 0  
5 22 C 0 100 50 50 0 0  
5 23 C 0 100 50 50 0 0  
5 24 C 0 100 50 50 0 0  
5 25 C 0 100 50 50 0 0  
5 26 C 0 100 50 50 0 0  
5 27 C 0 109 50 59 0 0  
5 28 C 0 110 60 50 0 0  
5 29 C 0 100 50 50 0 0  
5 30 C 0 120 50 70 0 0  
5 31 C 0 186 70 116 0 0  
5 32 C 0 110 50 60 0 0  
5 33 C 0 124 49 75 0 0  
5 34 C 0 110 60 50 0 0  
5 35 C 0 120 70 50 0 0  
5 36 C 0 125 65 60 0 0  
5 37 C 0 130 60 70 0 0  
5 38 C 0 135 70 65 0 0  
5 39 C 0 200 50 150 0 0  
5 40 C 0 125 65 60 0 0  
5 41 C 0 120 50 70 0 0  
5 42 C 0 100 50 50 0 0  
5 43 C 0 125 65 60 0 0  
5 44 C 0 124 65 59 0 0  
5 45 C 0 110 60 50 0 0  
5 46 C 0 125 60 65 0 0  
5 47 C 0 125 60 65 0 0  
5 48 C 1 140 80 60 0 1
```

```
5 49 C 0 110 50 60 0 0
5 50 C 0 110 50 60 0 0
5 51 C 0 196 116 80 0 0
5 52 C 0 110 60 50 0 0
5 53 C 0 120 70 50 0 0
5 54 C 0 120 60 60 0 0
5 55 C 0 135 65 70 0 0
5 56 C 3 120 60 60 3 0
5 57 C 0 120 55 65 0 0
5 58 C 2 60 0 60 0 2
5 59 C 0 55 0 55 0 0
6 60 C 0 160 80 80 0 0
6 61 C 1 159 80 79 0 1
6 62 C 1 159 79 80 1 0
6 64 C 0 160 80 80 0 0
6 65 C 0 160 80 80 0 0
6 66 C 0 160 80 80 0 0
6 67 C 0 160 80 80 0 0
6 68 C 0 159 80 79 0 0
6 69 C 0 160 80 80 0 0
6 70 C 0 160 80 80 0 0
6 71 C 0 184 92 92 0 0
;
```

```

/* power_prior.sas */

/* first run MM_data.sas */
/* creates MM_HCD_full */
/* input author_index study_index mouse_line $ N_tumors N_mice N_males
N_females N_male_tumors N_female_tumors; */

data MM_HCD;
    set MM_HCD_full;
    study_type = "HCD";
    trt = 0;
run;

data MM_base;
    input study_type $ trt N_mice N_tumors author_index study_index;
    datalines;
    ctr 0 43 0 7 72
    mut 1 93 2 7 73
    ;
run;

data MM_mice;
    set MM_HCD MM_base;
    keep author_index study_index study_type trt N_mice N_tumors;
run;

/*
proc print data = MM_mice noobs;
run;
*/

/* ----- */
/* log odds ratio, base only */

title "log odds ratio, base only";

proc mcmc data=MM_base nbi=1000 nmc=50000 outpost=postdata seed=11117;
    parm Pw Pm;
    prior Pw ~ beta(1,1);
    prior Pm ~ beta(1,1);

    if (trt = 0) then
        llike = logpdf("binomial", N_tumors, Pw, N_mice);
    else
        llike = logpdf("binomial", N_tumors, Pm, N_mice);

    model general(llike);
run;

data postdata;
    set postdata;
    OR = (Pm / (1- Pm )) / (Pw / (1- Pw));
    OR_ge1 = OR > 1;
    OR_ge2 = OR > 2;

```

```

run;

proc means data=postdata mean;
    var OR_ge1 OR_ge2;
run;

/* ----- */
/* log odds ratio, all */

title "log odds ratio, all HCD (lambda = 1) method 1";

proc mcmc data=MM_mice nbi=1000 nmc=50000 outpost=postdata seed=11117;
    parm Pw Pm;
    prior Pw ~ beta(1,1);
    prior Pm ~ beta(1,1);

    if (trt = 0) then
        llike = logpdf("binomial", N_tumors, Pw, N_mice);
    else
        llike = logpdf("binomial", N_tumors, Pm, N_mice);

    model general(llike);
run;

data postdata;
    set postdata;
    OR = (Pm / (1- Pm )) / (Pw / (1- Pw));
    OR_ge1 = OR > 1;
    OR_ge2 = OR > 2;
run;

proc means data=postdata mean;
    var OR_ge1 OR_ge2;
run;

title "log odds ratio, all HCD (lambda = 1) method 2";

proc mcmc data=MM_mice nbi=1000 nmc=50000 thin=5 outpost=postdata
seed=11117;
    parm a_int;
    parm theta;

    prior a_int ~ n(0,sd=100);
    prior theta ~ n(0,sd=25);

    p = logistic (a_int + theta*(trt-0.5));

    llike = logpdf("binomial", N_tumors, p, N_mice);

    model general(llike);
run;

data postdata;
    set postdata;

```

```

        OR_ge1 = (theta > 0); /* log(1)=0 */
        OR_ge2 = (theta >= log(2));
run;

proc means data=postdata mean;
    var OR_ge1 OR_ge2;
run;

/* ----- */
/* log odds ratio, power prior 0.3260 */

title "log odds ratio, power prior lambda_AK 0.3260 method 1";

proc mcmc data=MM_mice nbi=1000 nmc=50000 thin=5 outpost=postdata
seed=11117;
    parm Pw Pm;
    prior Pw ~ beta(1,1);
    prior Pm ~ beta(1,1);

    if (trt = 0) then do
        llike = logpdf("binomial", N_tumors, Pw, N_mice);
        if study_type = "HCD" then
            llike = 0.32604795399401865*llike;
        end;
    else
        llike = logpdf("binomial", N_tumors, Pm, N_mice);
    end;

    model general(llike);
run;

data postdata;
    set postdata;
    OR = (Pm / (1- Pm )) / (Pw / (1- Pw));
    OR_ge1 = OR > 1;
    OR_ge2 = OR > 2;
run;

proc means data=postdata mean;
    var OR_ge1 OR_ge2;
run;

title "log odds ratio, power prior lambda_AK 0.3260 method 2";

proc mcmc data=MM_mice nbi=1000 nmc=50000 thin=5 outpost=postdata
seed=11117;
    parm a_int;
    parm theta;

    prior a_int ~ n(0,sd=100);
    prior theta ~ n(0,sd=5);

    p = logistic (a_int + theta*(trt-0.5));

```

```

        llike = logpdf("binomial", N_tumors, p, N_mice);

        if study_type = "HCD" then
            llike = 0.32604795399401865*llike;

        model general(llike);
run;

data postdata;
    set postdata;
    OR_ge1 = (theta > 0); /* log(1)=0 */
    OR_ge2 = (theta >= log(2));
run;

proc means data=postdata mean;
    var OR_ge1 OR_ge2;
run;

/* ----- */
/* log odds ratio, power prior 0.2159 */

title "log odds ratio, power prior lambda_AKP 0.2159 method 1";

proc mcmc data=MM_mice nbi=1000 nmc=50000 thin=5 outpost=postdata
seed=11117;
    parm Pw Pm;
    prior Pw ~ beta(1,1);
    prior Pm ~ beta(1,1);

    if (trt = 0) then do
        llike = logpdf("binomial", N_tumors, Pw, N_mice);
        if study_type = "HCD" then
            llike = 0.21590848576693064*llike;
        end;
    else
        llike = logpdf("binomial", N_tumors, Pm, N_mice);

    model general(llike);
run;

data postdata;
    set postdata;
    OR = (Pm / (1- Pm )) / (Pw / (1- Pw));
    OR_ge1 = OR > 1;
    OR_ge2 = OR > 2;
run;

proc means data=postdata mean;
    var OR_ge1 OR_ge2;
run;

title "log odds ratio, power prior lambda_AKP 0.2159 method 2";

```

```

proc mcmc data=MM_mice nbi=1000 nmc=50000 thin=5 outpost=postdata
seed=11117;
  parm a_int;
  parm theta;

  prior a_int ~ n(0,sd=100);
  prior theta ~ n(0,sd=5);

  p = logistic (a_int + theta*(trt-0.5));

  llike = logpdf("binomial", N_tumors, p, N_mice);

  if study_type = "HCD" then
    llike = 0.21590848576693064*llike;

  model general(llike);
run;

data postdata;
  set postdata;
  OR_ge1 = (theta > 0); /* log(1)=0 */
  OR_ge2 = (theta >= log(2));
run;

proc means data=postdata mean;
  var OR_ge1 OR_ge2;
run;

```

```

/* betabinom_mcmc.sas */

/* first run MM_data.sas */
/* creates MM_HCD_full */
/* input author_index study_index mouse_line $ N_tumors N_mice N_males
N_females N_male_tumors N_female_tumors; */

data MM_HCD;
    set MM_HCD_full;
    study_type = "HCD";
    trt = 0;
run;

data MM_base;
    input study_type $ trt N_mice N_tumors author_index study_index;
    datalines;
    ctr 0 43 0 7 72
    mut 1 93 2 7 73
    ;
run;

data MM_mice;
    set MM_HCD MM_base;
    keep author_index study_index study_type trt N_mice N_tumors;
run;

* proc print data = MM_mice;
* run;

proc mcmc data=MM_mice nmc=20000 thin=2 outpost=postdata seed=11117;
    parm a_int;
    parm theta;
    parm psi;

    prior a_int ~ n(0,sd=100); /* logit (mu_j) = a_int + theta*(trt-
0.5)
                                j=ctr,mut; trt=0,1 for ctr,mut */

    prior theta ~ n(0,sd=5);
    prior psi ~ expon(scale=0.5);

    mu_j = logistic(a_int + theta*(trt-0.5)); /* the logistic
transformation of an argument;
                                used to convert a log odds value to a
value
                                on the probability scale. (SAS doc)
*/

    /* betabinomial logCDF x~betabin(n,alpha,beta) */
    x = N_tumors;
    n = N_mice;
    alpha = mu_j/psi;
    beta = (1-mu_j)/psi;

```

```

        log_nchx = lgamma(n+1) - lgamma(x+1) - lgamma(n-x+1);
        log_Bnum = lgamma(x+alpha) + lgamma(n-x+beta) -
lgamma(n+alpha+beta);
        log_Bdenom = lgamma(alpha) + lgamma(beta) - lgamma(alpha+beta);

        llike = log_nchx + log_Bnum - log_Bdenom;
        model general(llike);
run;

data postdata;
    set postdata;
    OR_ge1 = theta > 0;
    OR_ge2 = theta >= log(2);
run;

proc means data=postdata mean;
    var OR_ge1 OR_ge2;
run;

```